Pensieve header: Implementing KH, day 5.



```
Nice torus, Justin! But can you avoid the white cut? And the corners?
```

? PlotPoints

PlotPoints is an option for plotting functions that specifies how many initial sample points to use. >>



A nice pair of tori, Andrew! But I was not genuinely enlightened as for why S^4 is the union of two soild tori. Can you aim lower?

From http://www.math.toronto.edu/drorbn/classes/16-1750-ShamelessMathematica/About.html: **Possible Topics** (in no particular order). Whatever you may suggest, and the ~~Fibonacci numbers~~; ~~the Jones polynomial~~; ~~a more efficient Jones algorithm~~; ~~a riddle on spheres~~; ~~Khovanov homology~~; Γ-calculus; the Hopf fibration; Hilbert’s 13th problem; non-commutative Gaussian elimination; free Lie algebras; the Baker-Campbell-Hausdorff formula; wacky numbers; ~~an order 4 torus~~; the Schwarz Lantern; knot colourings; the Temperley-Lieb pairing; the dodecahedral link; sound experiments; barycentric subdivisions; a Peano curve; braid closures and Vogel’s algorithm; the insolubility of the quintic, phase portraits.

```
Import["http://www.math.toronto.edu/~drorbn/papers/Categorification/QRG.png"]
```

A Quick Reference Guide to Khovanov's Categorification of the Jones Polynomial

Dror Bar-Natan, June 12, 2002

The Kauffman Bracket: $\langle\emptyset\rangle = 1$; $\langle\bigcirc L\rangle = (q+q^{-1})\langle L\rangle$; $\langle\times\rangle = \langle \underset{0-\text{smoothing}}{\asymp} \rangle - q\langle \underset{1-\text{smoothing}}{)(} \rangle$.

The Jones Polynomial: $\hat{J}(L) = (-1)^{n_-}q^{n_+-2n_-}\langle L\rangle$, where (n_+, n_-) count $(\times, \times)$ crossings.

Khovanov's construction: $[\![L]\!]$ — a chain complex of graded $\mathbb{Z}$-modules;

$$[\![\emptyset]\!] = 0 \to \underset{\text{height } 0}{\mathbb{Z}} \to 0; \qquad [\![\bigcirc L]\!] = V \otimes [\![L]\!]; \qquad [\![\times]\!] = \text{Flatten}\left(0 \to \underset{\text{height } 0}{[\![\asymp]\!]} \to \underset{\text{height } 1}{[\![)(]\!]\{1\}} \to 0\right);$$

$$\mathcal{H}(L) = \mathcal{H}\left(\mathcal{C}(L) = [\![L]\!][-n_-]\{n_+ - 2n_-\}\right)$$

$$V = \text{span}\langle v_+, v_-\rangle; \qquad \deg v_\pm = \pm 1; \qquad q\text{dim}\, V = q + q^{-1} \quad \text{with} \quad q\text{dim}\,\mathcal{O} := \sum_m q^m \dim \mathcal{O}_m;$$

$$\mathcal{O}\{l\}_m := \mathcal{O}_{m-l} \quad \text{so} \quad q\text{dim}\,\mathcal{O}\{l\} = q^l\, q\text{dim}\,\mathcal{O}; \qquad \cdot[s]: \quad \text{height shift by } s;$$



$$\longrightarrow (V \otimes V \xrightarrow{m} V) \qquad m: \begin{cases} v_+ \otimes v_- \mapsto v_- & v_+ \otimes v_+ \mapsto v_+ \\ v_- \otimes v_+ \mapsto v_- & v_- \otimes v_- \mapsto 0 \end{cases}$$

$$\longrightarrow (V \xrightarrow{\Delta} V \otimes V) \qquad \Delta: \begin{cases} v_+ \mapsto v_+ \otimes v_- + v_- \otimes v_+ \\ v_- \mapsto v_- \otimes v_- \end{cases}$$

That's a Frobenius Algebra! And a (1+1)-dimensional TQFT!

Example: $\quad q^{-2} + 1 + q^2 - q^6 \xrightarrow[\text{(with } (n_+, n_-) = (3,0))]{\cdot(-1)^{n_-}q^{n_+-2n_-}} q + q^3 + q^5 - q^9.$



(here $(-1)^\xi := (-1)^{\sum_{i<j}\xi_i}$ if $\xi_j = \star$) $\qquad = \quad [\![K]\!] \xrightarrow[\text{(with } (n_+, n_-)=(3,0))]{\cdot[-n_-]\{n_+-2n_-\}} \mathcal{C}(K)$.

Theorem 1. The graded Euler characteristic of $\mathcal{C}(L)$ is $\hat{J}(L)$.

Theorem 2. The homology $\mathcal{H}(L)$ is a link invariant and thus so is $Kh_{\mathbb{F}}(L) := \sum_r t^r\, q\text{dim}\,\mathcal{H}^r_{\mathbb{F}}(\mathcal{C}(L))$ over any field $\mathbb{F}$.

Theorem 3. $\mathcal{H}(\mathcal{C}(L))$ is strictly stronger than $\hat{J}(L)$: $\mathcal{H}(\mathcal{C}(\bar{5}_1)) \neq \mathcal{H}(\mathcal{C}(10_{132}))$ whereas $\hat{J}(\bar{5}_1) = \hat{J}(10_{132})$.

Conjecture 1. $Kh_{\mathbb{Q}}(L) = q^{s-1}\left(1 + q^2 + (1+tq^4)Kh'\right)$ and $Kh_{\mathbb{F}_2}(L) = q^{s-1}(1+q^2)\left(1 + (1+tq^2)Kh'\right)$ for even $s = s(L)$ and non-negative-coefficients laurent polynomial $Kh' = Kh'(L)$.

Conjecture 2. For alternating knots s is the signature and Kh' depends only on tq^2.

References. Khovanov's arXiv:math.QA/9908171 and arXiv:math.QA/0103190 and DBN's
`http://www.ma.huji.ac.il/~drorbn/papers/Categorification/`.

Road map:

~~1. Produce all the info in a "cube of smoothings".~~

~~2. Produce a commuting cube of vector spaces and maps between them.~~

~~3. Produce a complex.~~

4. Compute homology.

5. Package everything nicely (almost) for public consumption.

```
K[3, 1] = Xm[1, 4, 2, 5] Xm[3, 6, 4, 1] Xm[5, 2, 6, 3];
K[5, 1] = Xm[1, 6, 2, 7] Xm[3, 8, 4, 9] Xm[5, 10, 6, 1] Xm[7, 2, 8, 3] Xm[9, 4, 10, 5];
K[10, 132] =
  Xm[5, 12, 6, 13] Xm[9, 16, 10, 17] Xm[11, 6, 12, 7] Xm[13, 20, 14, 1] Xm[15, 18, 16, 19]
   Xm[17, 10, 18, 11] Xm[19, 14, 20, 15] Xp[2, 8, 3, 7] Xp[4, 2, 5, 1] Xp[8, 4, 9, 3];
```

```
K = K[3, 1]
```

```
Xm[1, 4, 2, 5] Xm[3, 6, 4, 1] Xm[5, 2, 6, 3]
```

```
SetAttributes[{p, dx}, Orderless];
S[K_] := Module[{α = 0, ϵ, X, Z},
  ϵ /: ϵ^_ = 0;
  Z = Expand[
        dx[] K /. {
           (Xp | Xm)[i_, j_, k_, l_] :> (++α;
             p[i, j] p[k, l] + dx[α] p[i, l] p[j, k] + ϵ e[α] X[i, j, k, l])
          } /. p[i_, j_] :> p[i, j][Min[i, j]]
       ] //. p[i_, j_][m_] p[j_, k_][n_] :> p[i, k][Min[m, n]] /. {
       X[i_, j_, k_, l_] p[i_, j_][m_] p[k_, l_][n_] :> (c[m] c[n] → c[Min[m, n]]),
       X[i_, j_, k_, l_] p[i_, l_][m_] p[j_, k_][n_] :> (c[Min[m, n]] → c[m] c[n])
      } /.
     p[i_, j_][m_]^_. :> c[m] //. dx[i___] dx[j___] :> dx[i, j];
  CoefficientList[Z, ϵ, {2}]
 ]
```

```
({S0, S1} = S[K]) // Column
```

```
c[1] c[2] c[3] dx[] + c[1] c[3] dx[1] + c[1] c[2] dx[2] + c[1] c[2] dx[3] +
 c[1] dx[1, 2] + c[1] dx[1, 3] + c[1] dx[2, 3] + c[1] c[2] dx[1, 2, 3]
dx[2, 3] e[1] (c[1] → c[1] c[2]) + dx[1, 3] e[2] (c[1] → c[1] c[2]) +
 dx[1, 2] e[3] (c[1] → c[1] c[2]) + c[3] dx[] e[1] (c[1] c[2] → c[1]) +
 dx[2] e[1] (c[1] c[2] → c[1]) + dx[3] e[1] (c[1] c[2] → c[1]) +
 dx[3] e[2] (c[1] c[2] → c[1]) + dx[2] e[3] (c[1] c[2] → c[1]) +
 c[2] dx[] e[2] (c[1] c[3] → c[1]) + dx[1] e[2] (c[1] c[3] → c[1]) +
 dx[1] e[3] (c[1] c[3] → c[1]) + c[1] dx[] e[3] (c[2] c[3] → c[2])
```

```
V0 = List @@ Expand[S0 /. c[i_] :> vp[i] + vm[i]]
```

```
{dx[1, 2] vm[1], dx[1, 3] vm[1], dx[2, 3] vm[1], dx[2] vm[1] vm[2],
 dx[3] vm[1] vm[2], dx[1, 2, 3] vm[1] vm[2], dx[1] vm[1] vm[3],
 dx[] vm[1] vm[2] vm[3], dx[1, 2] vp[1], dx[1, 3] vp[1], dx[2, 3] vp[1],
 dx[2] vm[2] vp[1], dx[3] vm[2] vp[1], dx[1, 2, 3] vm[2] vp[1], dx[1] vm[3] vp[1],
 dx[] vm[2] vm[3] vp[1], dx[2] vm[1] vp[2], dx[3] vm[1] vp[2], dx[1, 2, 3] vm[1] vp[2],
 dx[] vm[1] vm[3] vp[2], dx[2] vp[1] vp[2], dx[3] vp[1] vp[2], dx[1, 2, 3] vp[1] vp[2],
 dx[] vm[3] vp[1] vp[2], dx[1] vm[1] vp[3], dx[] vm[1] vm[2] vp[3], dx[1] vp[1] vp[3],
 dx[] vm[2] vp[1] vp[3], dx[] vm[1] vp[2] vp[3], dx[] vp[1] vp[2] vp[3]}
```

```
sign[αs_List, β_Integer] := (-1)^Count[αs, α_ /; α<β]
```

```
d[dx[αs___] e[β_] cs_. (c[i_] c[j_] → c[k_])] := {s = sign[{αs}, β];
    dx[αs] vp[i] vp[j] → s dx[αs, β] vp[k],
    dx[αs] vm[i] vp[j] → s dx[αs, β] vm[k],
    dx[αs] vp[i] vm[j] → s dx[αs, β] vm[k],
    dx[αs] vm[i] vm[j] → 0,
    dx[βs___] /; {βs} =!= {αs} :> 0
   };
d[dx[αs___] e[β_] cs_. (c[k_] → c[i_] c[j_])] := {s = sign[{αs}, β];
   dx[αs] vp[k] → s dx[αs, β] (vp[i] vm[j] + vm[i] vp[j]),
   dx[αs] vm[k] → s dx[αs, β] vm[i] vm[j],
   dx[βs___] /; {βs} =!= {αs} :> 0
  }
```

To do: ~~1. Signs.~~ ~~2. Deg.~~ ~~3. C0[r,deg].~~ ~~4. d0[r].~~ ~~4.5. Verify d0//d0==0.~~ 5. rankr=r0[r,deg]. 6. Put it all together!

```
Deg[dx[αs___] rest_] :=
 Length@{αs} + Count[rest, _vp, {0, 1}] - Count[rest, _vm, {0, 1}]
```

```
C0[r_, deg_] := Cases[V0, as_dx rest_ /; Length@as == r ∧ Deg[as rest] == deg];
C0[r_] := Cases[V0, as_dx rest_ /; Length@as == r];
```

```
C0[2, 1]
```

```
{dx[1, 2] vm[1], dx[1, 3] vm[1], dx[2, 3] vm[1]}
```

```
r = 2
```

```
2
```

```
Dim[dx[αs___] rest_] := Length@{αs} ;
d0[r_][expr_] := Plus @@ (expr /. (d /@ Select[List @@ S1, Dim[#] == r &]))
```

```
C0[1]

{dx[2] vm[1] vm[2], dx[3] vm[1] vm[2], dx[1] vm[1] vm[3], dx[2] vm[2] vp[1],
 dx[3] vm[2] vp[1], dx[1] vm[3] vp[1], dx[2] vm[1] vp[2], dx[3] vm[1] vp[2],
 dx[2] vp[1] vp[2], dx[3] vp[1] vp[2], dx[1] vm[1] vp[3], dx[1] vp[1] vp[3]}

d0[1] /@ C0[1]

{0, 0, 0, dx[1, 2] vm[1] - dx[2, 3] vm[1],
 dx[1, 3] vm[1] + dx[2, 3] vm[1], -dx[1, 2] vm[1] - dx[1, 3] vm[1],
 dx[1, 2] vm[1] - dx[2, 3] vm[1], dx[1, 3] vm[1] + dx[2, 3] vm[1],
 dx[1, 2] vp[1] - dx[2, 3] vp[1], dx[1, 3] vp[1] + dx[2, 3] vp[1],
 -dx[1, 2] vm[1] - dx[1, 3] vm[1], -dx[1, 2] vp[1] - dx[1, 3] vp[1]}

d0[2] /@ (d0[1] /@ C0[1])

{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

{r = 2, deg = 3}

{2, 3}

Expand[d0[r] /@ C0[r, deg]]

{dx[1, 2, 3] vm[2] vp[1] + dx[1, 2, 3] vm[1] vp[2],
 -dx[1, 2, 3] vm[2] vp[1] - dx[1, 2, 3] vm[1] vp[2],
 dx[1, 2, 3] vm[2] vp[1] + dx[1, 2, 3] vm[1] vp[2]}

C0[r + 1, deg]

{dx[1, 2, 3] vm[2] vp[1], dx[1, 2, 3] vm[1] vp[2]}

? Coefficient
```

Coefficient[*expr*, *form*] gives the coefficient of *form* in the polynomial *expr*.

Coefficient[*expr*, *form*, *n*] gives the coefficient of *form* ^ *n* in *expr*. >>

```
mat = Table[
   Coefficient[Expand[d0[r][x]], y],
   {x, C0[r, deg]}, {y, C0[r + 1, deg]}
  ];
mat // MatrixForm
```

$$\begin{pmatrix} 1 & 1 \\ -1 & -1 \\ 1 & 1 \end{pmatrix}$$

```
MatrixRank[mat]

1
```

```
r0[r_, deg_] := (
  bas0 = C0[r, deg];
  bas1 = C0[r + 1, deg];
  If[bas0 == {} ∨ bas1 == {}, 0,
   mat = Table[
     Coefficient[Expand[d0[r][x]], y],
     {x, bas0}, {y, bas1}
    ];
   MatrixRank[mat]
  ]
 )
```

```
r0[2, 3]
```

1

```
r0[2, 4]
```

0

```
β0[r_, deg_] := Length@C0[r, deg] - r0[r, deg] - r0[r - 1, deg]
```

```
β0[2, 3]
```

0

```
Union[Deg /@ V0]
```

{-3, -1, 1, 3, 5}

```
Table[β0[r, deg],
 {r, 0, Length[K]}, {deg, Union[Deg /@ V0]}
]
```

{{1, 0, 0, 0, 0}, {0, 0, 1, 0, 0}, {0, 0, 0, 0, 0}, {0, 0, 0, 1, 1}}

```
Kh0 = Sum[t^r q^deg β0[r, deg],
  {r, 0, Length[K]}, {deg, Union[Deg /@ V0]}
 ]
```

$\frac{1}{q^3} + q\, t + q^3 t^3 + q^5 t^3$

```
J0 = Kh0 /. t → -1
```

$\frac{1}{q^3} - q - q^3 - q^5$

```
Expand[-q^6 (J0 /. q → 1/q)]
```

$q + q^3 + q^5 - q^9$